

Matemàtiques amb ordinador.

Curs pràctic de Maple

Àlgebra lineal

Curs 2003–04

Índex

II	Àlgebra lineal amb <code>LinearAlgebra</code>	65
9	Introducció al paquet <code>LinearAlgebra</code>	65
9.1	Introducció de les dades.	65
9.2	Les operacions entre matrius.	66
9.3	La matriu d'un sistema d'equacions lineals.	68
9.4	Les operacions de reducció.	69
9.5	Reduccions automàtiques.	72
9.6	Càlcul del rang	73
9.7	Bases d'espais vectorials	75
10	Endomorfismes, vectors i valors propis, diagonalització. . .	79
10.1	Polinomi característic i valors propis	79
10.2	Vectors propis	79
10.3	Formes de Jordan	82
11	Càlculs de forma interactiva (<i>maplets</i>)	85
11.1	<code>GaussianEliminationTutor</code>	85
11.2	<code>InverseTutor</code>	86
11.3	<code>EigenvectorsTutor</code>	86

Part II

Àlgebra lineal amb **LinearAlgebra**

9 Introducció al paquet **LinearAlgebra**

Quan es treballa en problemes que tenen a veure amb l'àlgebra lineal els càlculs que s'han de realitzar consisteixen, bàsicament, en la manipulació de matrius i sovint el volum de dades amb les que s'ha de treballar fa que les operacions més simples es converteixin en feines pesades i subjectes a errors incontrolables.

El paquet de Maple **LinearAlgebra** permet treballar amb matrius d'una forma còmoda i ràpida. Teniu en compte que no és l'únic que existeix per a realitzar aquest tipus de càlculs (per exemple, en la instal·lació normal de Maple també trobareu el paquet **linalg**) i les mateixes operacions es poden realitzar amb instruccions de nom i sintaxi diferents per a cada paquet. A continuació presentem les que són més fonamentals per a treballar amb problemes típics d'àlgebra lineal utilitzant **LinearAlgebra**.

Com sempre, comenceu netejant la memòria de Maple i *carregant* el paquet amb el que treballareu:

```
> restart;  
> with(LinearAlgebra):
```

9.1 Introducció de les dades.

Els objectes sobre els que treballen les funcions del paquet **LinearAlgebra** són les matrius i els vectors. Per a introduir una matriu o un vector podem utilitzar la notació $\langle \ , \ , \dots \rangle$ o $\langle \ | \ | \ | \dots \rangle$ segons si volem fer referència a les files de l'objecte (notació amb comes) o a les columnes (notació amb barres).

Exemple 9.1

La matriu $M = \begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{pmatrix}$ es pot introduir per files com

```
> Mf:=< < 1 | 2 | 3 > , < 4 | 5 | 6 > >;
```

o per columnes com

```
> Mc:= < < 1 , 4 > | < 2 , 5 > | < 3 , 6 > >;
```

Fixeu-vos que en els dos casos es veu el mateix resultat exactament.

Aquestes construccions només són formes abreujades i simplificades d'introduir les matrius o vectors, podeu consultar a l'ajuda del programa les funcions per a escriure les matrius i els vectors que són **Matrix()** i **Vector()** i podreu veure totes les possibilitats que tenen. Podeu veure en l'exemple següent alguns casos particulars de d'introducció de matrius i vectors utilitzant aquestes comandes. En particular, es pot veure com es pot utilitzar una funció que tingui com arguments els índexs de cada una de les posicions d'una matriu per a calcular els seus coeficients.

Exemple 9.2

```

> M:=Matrix([[1,-2],[0,1]]);
> v:=Vector([2,3]);
> w:=Vector[row]([2,3]);
> f:=(i,j)->x^i*y^j;
> Nf:=Matrix(2,f);
> N:=subs(x=3,y=2,Nf);

```

Per a extreure una submatriu d'una matriu ja definida podem utilitzar la sintaxi usual a les llistes. Per exemple, si A és una matriu, $A[i,j]$ és el coeficient de la fila i , columna j . Tenint en compte que tant l'argument i com el j poden ser rangs i llistes.

Exemple 9.3

Considerem la matriu Mf de l'exemple 9.1 i proveu:

```

> Mf[2,1];
> Mf[1..2,2..3];
> Mf[[2,1],1..3];

```

També heu de saber que el paquet **LinearAlgebra** té funcions per a introduir molts tipus especials de matrius amb poc esforç (matrius diagonals, triangulars superiors, amb tots els coeficients iguals,...). Pot ser ara és un bon moment per a fer un cop d'ull al joc de funcions d'aquest paquet mirant la secció de l'ajuda del programa dedicada al paquet **LinearAlgebra** on hi trobareu la llista completa.

```

> ?LinearAlgebra

```

9.2 Les operacions entre matrius.

Les operacions bàsiques entre matrius són la suma (i la diferència), el producte per un escalar i el producte de matrius. Quan utilitzem el paquet **LinearAlgebra** totes aquestes operacions estan disponibles de forma immediata. Podeu comprovar-ho en els exemples següents.

Exemple 9.4

```

> M1:= < <1,2,-1>|<0,3,3>|<-1,1,-3> >;
> M2:= < <0,1,-1>|<-2,3,5>|<1,1,0> >;
> M1+M2;
> M1-M2;
> 3*M1;
> 2*M1-4*M2;

```

Només cal tenir en compte que, com que el producte de matrius no es fa multiplicant cada una de les components dels factors, l'operador * (asterisc) no produeix el resultat que esperariem en un principi (de fet dóna un missatge d'error). Per a obtenir el vertader producte de matrius cal utilitzar l'operador . (punt).

Exemple 9.5

```
> M1.M2;
> M1*M2;
> M2.M1;
```

El producte de matrius no sempre es pot efectuar. Si les matrius que intenteu multiplicar no són de les mides correctes, Maple us donarà un missatge d'error.

Exemple 9.6

```
> M3:= < <1,2>|<2,-1>|<1,-1> >;
> M1.M3;
En canvi sí que es pot realitzar l'operació amb l'ordre invers.
> M3.M1;
```

Per a practicar una mica més amb la introducció de dades en vectors i matrius i en la realització dels càlculs elementals amb matrius aquí teniu un petit exercici.

Exercici 9.1

Introduïu la matriu $P_1 = \begin{pmatrix} \frac{4}{7} & \frac{1}{7} & -\frac{1}{7} \\ -\frac{5}{21} & \frac{4}{21} & \frac{1}{7} \\ -\frac{3}{7} & \frac{1}{7} & -\frac{1}{7} \end{pmatrix}$, el vector $v_1 = \begin{pmatrix} 1 \\ -\frac{11}{3} \\ -2 \end{pmatrix}$ i el vector $v_2 = \begin{pmatrix} 1 \\ -1 \\ 2 + 2\sqrt{2} \end{pmatrix}$.

Utilitzant la matriu $M_1 = \begin{pmatrix} 1 & 0 & -1 \\ 2 & 3 & 1 \\ -1 & 3 & -3 \end{pmatrix}$, que ja hauríeu de tenir introduïda en l'exemple 9.2

d'aquest apartat, calculeu els productes $(M_1) \cdot (P_1)$, $(M_1) \cdot (v_i)$, $(P_1) \cdot (v_i)$ (per a $i = 1$ i 2). Quina relació creieu que hi ha entre els productes $(M_1) \cdot (v_i)$, $(P_1) \cdot (v_i)$ i els vectors v_i ? Què passa si intenteu realitzar el producte $(v_1) \cdot (M_1)$? Busqueu en l'ajuda de Maple com es pot transposar un vector o una matriu i calculeu el producte del vector fila amb les mateixes components que v_1 per la matriu M_1 .

9.3 La matriu d'un sistema d'equacions lineals.

Quan es vol discutir i resoldre un sistema d'equacions lineals, el procediment usual consisteix en obtenir un sistema equivalent (és a dir, amb el mateix conjunt de solucions) que tingui una forma *reduïda*. (El significat de reduït depèn del context. En general podem dir que un sistema és reduït si podem dir sense cap mena de dubte si té solucions o no i en el cas que en tingui puguem llegir, sense fer càlculs addicionals, quines són aquestes solucions). Aquest procediment de *reducció* sempre es podrà realitzar utilitzant tres tipus bàsics de transformacions:

- Intercanviar dues de les equacions entre si.
- Multiplicar una de les equacions per una constant no nul·la.
- Sumar a una de les equacions un múltiple d'una de les altres.

Recordeu que per a poder fer aquestes operacions només necessitem treballar amb els *coeficients* de les equacions, sense utilitzar per res les *incògnites*. Així, per exemple, si es vol treballar sobre

el sistema d'equacions
$$\left. \begin{array}{l} 3x - 2y = 2 \\ 2x - y = 7 \end{array} \right\}$$
 bastarà que treballem sobre la matriu $\begin{pmatrix} 3 & -2 & 2 \\ 2 & -1 & 7 \end{pmatrix}$.

Maple ens permet fer aquest procés de forma automàtica. Per això el paquet **LinearAlgebra** proporciona la comanda `GenerateMatrix()` (i la comanda inversa `GenerateEquations()`).

Exemple 9.7

Per a obtenir la matriu de coeficients i el terme independent del sistema d'equacions

$$\left. \begin{array}{l} x - y + z - 2t = 1 \\ y - 2z = 6 \\ 2x + 4y + t = 8 \\ 44x - y + 34z - 21t = 98 \end{array} \right\}$$

posarem

```
> sistema1:=[x-y+z-2*t=1, y-2*z=6, 2*x+4*y+t=8,44*x-y+34*z-21*t=98];
> vars:=[x,y,z,t];
> S1:= GenerateMatrix(sistema1,vars);
```

Noteu que `S1` és una llista de dues matrius, la primera amb els coeficients i l'altra amb el terme independent. Si el que voleu és la matriu ampliada del sistema podeu fer

```
> Sa1:= <S1[1]|S1[2]>;
```

o bé utilitzar l'opció de la comanda `GenerateMatrix()` que fa el procés anterior d'un sol cop

```
> GenerateMatrix(sistema1,vars,augmented);
```

Noteu que és molt important especificar quines són les incògnites ja que, si no es fa, Maple no seria capaç de distingir en un sistema d'equacions lineals que depèn d'un paràmetre si les indeterminades que hi surten són totes incògnites o no.

Exemple 9.8

Si tenim el sistema d'equacions, dependent del paràmetre λ , donat per $\left. \begin{array}{l} x - 2y = \lambda \\ 2x - 3y = 2 - \lambda \end{array} \right\}$, per a obtenir la seva matriu de coeficients haurem de fer

```
> GenerateMatrix([x-2*y=lambda,2*x-3*y=2-lambda],[x,y]);
```

encara que dins d'aquest sistema la dependència respecte λ també sigui lineal. Si no posem quines són les incògnites Maple donarà un missatge d'error.

```
> GenerateMatrix([x-2*y=lambda,2*x-3*y=2-lambda]);
```

L'exemple següent mostra el procediment per a generar un sistema d'equacions a partir de la matriu de coeficients utilitzant `GenerateEquations()`. Els detalls sobre aquesta comanda els podreu trobar a l'ajuda de Maple en l'apartat corresponent a la funció `GenerateEquations` del paquet **LinearAlgebra**.

Exemple 9.9

El sistema d'equacions lineals que té per matriu de coeficients $\begin{pmatrix} 1 & -2 & 3 \\ 2 & 3 & 1 \\ -1 & 1 & -5 \end{pmatrix}$ i per terme

independent $\begin{pmatrix} 2 \\ -1 \\ 7 \end{pmatrix}$ respecte les incògnites $\begin{pmatrix} x \\ y \\ z \end{pmatrix}$ s'obtindrà amb la comanda:

```
> eqs:=GenerateEquations(<<1,2,-1>|<-2,3,1>|<3,1,-5>>,[x,y,z],<2,-1,7>);
```

El resultat serà una llista amb les tres equacions que volíem, de la que podem extreure cada un dels seus elements amb la notació usual `eqs[i]` per a i prenent valors entre 1 i 3.

9.4 Les operacions de reducció.

La comanda de Maple `RowOperation()` permet realitzar **totes** les operacions de reducció de la matriu d'un sistema d'equacions lineals. En els exemples que venen tot seguit podeu veure una

mostra de cada un dels tipus aplicat sobre la matriu $\begin{pmatrix} 2 & 1 & -1 \\ 3 & 5 & -2 \\ 4 & 8 & 16 \end{pmatrix}$.

```
> matriu:= < <2|1|-1>,<3|5|-2>,<4|8|16>>;
```

Exemple 9.10

Primer intercanviem la primera fila amb la tercera:

```
> RowOperation(matriu, [1, 3]);
```

Fixeu-vos que els arguments són: la matriu sobre la que volem actuar i una llista amb el parell files que voleu intercanviar.

Exemple 9.11

També podem multiplicar tota la tercera fila per $\frac{1}{4}$:

```
> RowOperation(matriu, 3, 1/4);
```

Aquí heu de notar que els arguments són: la matriu sobre la que volem actuar, la fila sobre la que volem fer la operació i finalment el nombre pel que volem multiplicar.

Exemple 9.12

I finalment podem sumar a la segona fila la primera multiplicada per $\frac{-3}{2}$ (cosa que *elimina* la primera incògnita en la segona equació).

```
> RowOperation(matriu, [2, 1], -3/2);
```

En aquest cas els arguments són: la matriu sobre la que actuem, una llista amb dos nombres que designen, el primer, la fila sobre la que volem realitzar l'operació i, el segon, la fila que utilitzem per actuar sobre la primera i, com a últim argument, el nombre pel que volem multiplicar.

Cal dir que en tots aquests casos també es podria afegir un argument opcional (*inplace*) que té l'efecte de realitzar les operacions i modificar la mateixa matriu sobre la que les fem d'acord amb el resultat obtingut (fixeu-vos que utilitzar aquesta opció té el perill d'equivocar-se amb l'operació que es vol realitzar i no saber, o poder, desfer el desastre i l'avantatge de que després d'executar la comanda ja sabem que la matriu a quedat modificada d'acord amb les operacions que es volien fer).

Exemple 9.13

Podem modificar *matriu* per a deixar una matriu reduïda realitzant les operacions següents:


```

> matriu;
> RowOperation(matriu, [2,1], -3/2, inplace);
> RowOperation(matriu, [3,1], -2, inplace);
> RowOperation(matriu, [3,2], -12/7, inplace);
> RowOperation(matriu, 1, 1/2, inplace);
> RowOperation(matriu, 2, 2/7, inplace);
> RowOperation(matriu, 3, 7/132, inplace);
> matriu;

```

També es pot dir que existeix la comanda `ColumnOperation()` amb les mateixes característiques que la comanda de les operacions per files però aplicant les operacions a les columnes.

Ara ja podeu mirar de fer els següents exercicis:

Exercici 9.2

Donat el sistema d'equacions

$$\left. \begin{array}{l} 2x_1 - 4x_2 + 3x_3 - 5x_4 + x_5 = 3 \\ x_1 - 4x_2 - 2x_3 - 7x_5 = 2 \\ 5x_1 - 3x_3 - 2x_4 + 2x_5 = 1 \\ x_2 - 4x_3 + 3x_4 - x_5 = \frac{3}{2} \end{array} \right\} \text{ genereu la seva matriu ampliada,}$$

realitzeu les operacions de reducció necessàries en aquesta matriu i determineu si té solucions o no en té. Si en té, com són aquestes solucions?

Exercici 9.3

En la fabricació d'un cert producte alimentari s'utilitzen quatre components diferents. Una unitat de cada un dels ingredients conté les quantitats de vitamines (miligramms) i de kilocalories que s'expressen en la taula següent:

	Comp1	Comp2	Comp3	Comp4
vit A	1	1	1	2
vit B	1	2	1	3
vit C	1	3	2	1
Kilocal	2	2	1	1

Quines quantitats de cada un dels ingredients s'han de barrejar per a produir un producte final que aportí, exactament, 300 mg de vitamina A, 430 mg de vitamina B, 310 mg de vitamina C i tingui un valor energètic de 250 kilocalories?

És possible fabricar un producte final en el que la combinació de vitamines i d'aportació energètica sigui qualsevol?

Exercici 9.4

Estudieu, per als diferents valors dels paràmetres b i t , el sistema d'equacions lineals respecte x , y , z .

$$\left. \begin{aligned} 3x + 2y + z &= t \\ x - y + 2z &= 1 + t^2 \\ 3x + 7y - 4z &= -1 - t - t^2 - t^3 \\ 2x + y + bz &= t^3 \end{aligned} \right\}$$

9.5 Reduccions automàtiques.

Sovint no serà necessari realitzar les operacions de reducció sobre la matriu d'un sistema d'equacions lineals una per una. La comanda de Maple `Pivot()` és en aquests casos força útil. Quan executem la comanda `Pivot(A, i, j)`, se sumen a totes les files de la matriu A múltiples adequats de la fila que ocupa la posició i per tal que els coeficients en la columna j siguin tots nuls (naturalment la fila i no es modifica, tampoc es pot realitzar la operació si el coeficient en la columna j de la fila i és 0). Aquesta comanda també té l'opció `inplace` per a realitzar els càlculs modificant la matriu sobre la que estem treballant.

Exemple 9.14

Si es considera la matriu donada per $A = \begin{pmatrix} 1 & 3 & 0 \\ 2 & -1 & -5 \\ 0 & 1 & 3 \end{pmatrix}$

```
> A:= <<1|3|0>,<2|-1|-5>,<0|1|3>>;
```

Podem *pivotar* sobre l'element $[3, 2]$ amb

```
> Pivot(A,3,2);
```

Però no ho podrem fer sobre l'element $[3, 1]$.

```
> Pivot(A,3,1);
```

Quan no hi ha cap indeterminació en el camí per a fer la reducció d'un sistema d'equacions lineals es pot demanar a Maple que redueixi d'un sol cop la matriu del sistema. La comanda de Maple `GaussianElimination()` és la que realitza la reducció *Gaussiana* (dóna com a resultat una matriu reduïda amb zeros a la part inferior)

Exemple 9.15

Apliquem `GaussianElimination()` sobre la matriu $A = \begin{pmatrix} 1 & 3 & 0 \\ 2 & -1 & -5 \\ 0 & 1 & 3 \end{pmatrix}$ (que és la de l'exemple anterior).

```
> GaussianElimination(A);
```

Per a reduir completament la matriu del sistema d'equacions lineals es pot utilitzar la comanda `ReducedRowEchelonForm()` que a part de la reducció Gaussiana torna com a resultat una matriu en la que els *pivots* són 1 i tots els altres coeficients de la columna d'un d'aquests pivots són 0.

```
> ReducedRowEchelonForm(A);
```

Finalment, si el que es vol és determinar la solució d'un sistema d'equacions lineals donat per la seva matriu i pel terme independent, es pot utilitzar la comanda `LinearSolve()`.

Exemple 9.16

La solució del sistema d'equacions amb matriu de coeficients A i terme independent $\begin{pmatrix} 1 \\ 2 \\ -5 \end{pmatrix}$ es pot obtenir amb

```
> LinearSolve(A,<1,2,-5>);
```

Encara que el sistema sigui indeterminat, la comanda `LinearSolve()` també funciona i

```
> LinearSolve(<<1|2|1>,<-2|1|-5>>,<2,-1>);
```

dóna el conjunt de totes les solucions possibles en funció dels paràmetres necessaris.

9.6 Càlcul del rang

Una de les formes d'obtenir el rang d'una matriu consisteix en determinar-ne la submatriu més gran que conté amb determinant no nul. Podem calcular el determinant d'una matriu amb la comanda `Determinant`.

Recordeu també que tenir determinant diferent de 0 és la condició que ha de complir una matriu per a tenir una *matriu inversa*. La matriu inversa d'una matriu A es pot calcular amb la comanda `MatrixInverse(A)` (i també podem escriure $1/A$ o A^{-1}).

Exemple 9.17

```

> M:=<<1,-2>|<0,1>>;
> d:=Determinant(M);
> IM:=MatrixInverse(M);
> IM.M; M.IM;
> 1/M; M^(-1);

```

Exercici 9.5

Realitzeu un procediment `dd` que tingui com a únic argument un nombre natural n i calculi el determinant d'una matriu $n \times n$ que conté x 's a la diagonal i y 's fora de la diagonal (noteu que l'únic que cal fer és construir la matriu i aplicar la funció `Determinant`).

Estudieu per a quines combinacions de x i de y la matriu resultant és invertible o no. (Us serà útil utilitzar la comanda `factor` per a descomposar l'expressió del determinant).

Si l'únic que interessa és la determinació del rang d'una matriu A , la comanda `Rank(A)` és la que ens retorna aquest nombre.

Exemple 9.18

```

> M:=<<1,-2>|<0,1>>; N:=<<1,-2>|<2,-4>>;
> Rank(M);
> Rank(N);

```

Noteu que la funció `Rank` pot donar resultats incorrectes si entre els coeficients de la matriu hi ha paràmetres indeterminats i que, per tant, hi ha situacions en les que el càlcul no és pot fer directament. Per exemple

Exemple 9.19

Si s'aplica `Rank` a la matriu $\begin{pmatrix} x & 1 \\ 1 & x \end{pmatrix}$ el resultat serà 2 mentre que és clar que per a $x = 1$ o $x = -1$ el rang és 1.

Exercici 9.6

Determineu el rang de la matriu següent

$$\begin{pmatrix} 4-x & 2 & -2+\frac{2}{3}x & 8-x & -4 \\ \frac{13}{2} & -1 & x-3 & 10 & -9/2 \\ 4x-4 & 6 & 2-\frac{4}{3}x & 3x-8 & 4 \\ 6x-4 & 3 & 2-2x & 3x-8 & 4+x \end{pmatrix}$$

en funció del paràmetre x .

Si E és un subespai vectorial generat per vectors $\{v_1, v_2, \dots, v_n\}$, la dimensió del subespai vectorial E és el rang de la matriu que conté els vectors generadors a les files (o a les columnes). D'aquesta manera utilitzant la comanda **Rank** podem calcular la dimensió d'un subespai vectorial donada una família de vectors generadors.

Exercici 9.7

Calculeu la dimensió del subespai vectorial de \mathbb{R}^4 generat pels vectors $v_1 = (1, 3, -4, 5)$, $v_2 = (0, 6, 5, -1)$ i $v_3 = (1, -9, -14, 7)$.

9.7 Bases d'espais vectorials

Donada una matriu, Maple no només en sap calcular el rang sinó que a més també ens calcula una base del subespai generat tant pels vectors files com columnes. Les comandes que proporcionen aquesta informació són **RowSpace** i **ColumnSpace**. Observeu el seu funcionament en el exemple següent.

Exemple 9.20

```
> M:=<<1,-2,5,6>|<0,1,7,3>|<1,-4,-9,0>>;
> ColumnSpace(M);
> RowSpace(N);
```

Aleshores, la solució de l'exercici següent serà immediata si s'utilitza una d'aquestes comandes.

Exercici 9.8

Calculeu una base del subespai vectorial E de \mathbb{R}^4 generat pels vectors $v_1 = (1, 3, -4, 5)$, $v_2 = (0, 6, 5, -1)$ i $v_3 = (1, -9, -14, 7)$.

Quan es treballa amb diferents subespais vectorials (cada un d'ells determinat per un sistema de generadors) un problema que apareix sovint és la determinació d'una base de la suma o de la intersecció d'aquests espais. En el paquet **LinearAlgebra** hi ha les comandes **SumBasis** i **IntersectionBasis** que fan aquesta feina. Mireu quina és la sintaxi d'aquestes funcions i resoleu l'exercici següent.

Exercici 9.9

Sigui E el subespai vectorial de l'exercici anterior i F el subespai generat per $w_1 = (1, -6, -18, 12)$ i $w_2 = (3, 0, 0, 1)$. Calculeu la dimensió i bases per E , F , $E + F$ i $E \cap F$.

Noteu que quan apareixen paràmetres en el problema tot és una mica més delicat resolent aquest altre exercici.

Exercici 9.10

Sigui $E = \langle (4 - \lambda), 2, 1, (2 - \lambda), 4, 2 \rangle$ i $F = \langle (2 - \lambda), 4, 8 \rangle$. Determineu les dimensions de E , F , $E + F$ i $E \cap F$ segons el paràmetre λ .

Donada una matriu M , Maple també ens permet obtenir una base pel subespai vectorial de solucions del corresponent sistema homogeni mitjançant la comanda **NullSpace**. En l'exemple següent podeu veure com s'obté una base del subespai de \mathbb{R}^4 format pels vectors (x, y, z, t) tals que $x - 2y + 5z + 6t = 0$, $y + 7z + 3t = 0$, i $x - 4y - 9z = 0$.

Exemple 9.21

```
> M:=<<1,-2,5,6>|<0,1,7,3>|<1,-4,-9,0>>;
> NullSpace(M);
```

També és ben conegut que qualsevol subespai de \mathbb{R}^n coincideix amb l'espai de solucions d'un determinat sistema homogeni. En l'exemple següent podeu veure una manera d'obtenir aquest

sistema a partir d'un conjunt de generadors del subespai.

Exemple 9.22

Considerem el subespai de \mathbb{R}^5 generat pels vectors $(1, 2, 2, -1, 0)$, $(0, 1, -1, 2, 3)$ i $(2, -1, -4, 5, 1)$. Definim la matriu M com la que té per columnes les components dels tres vectors donats

$$> M := \langle \langle 1, 2, 2, -1, 0 \rangle \mid \langle 0, 1, -1, 2, 3 \rangle \mid \langle 2, -1, -4, 5, 1 \rangle \rangle;$$

afegim una columna al final que representa les components d'un vector qualsevol de \mathbb{R}^5

$$> Ma := \langle M \mid \langle x_1, x_2, x_3, x_4, x_5 \rangle \rangle;$$

realitzem les operacions de reducció per files fins a arribar a una forma reduïda. Les equacions en x_1, x_2, x_3, x_4, x_5 corresponents a files nul·les donen un sistema homogeni que té com espai de solucions l'espai generat pels vectors que hem considerat al principi. Donarà un resultat equivalent al sistema $13x_4 - 3x_1 - 9x_2 + 17x_3 = 0$, $13x_5 - 23x_2 + 14x_1 + 16x_3 = 0$.

10 Endomorfismes, vectors i valors propis, diagonalització...

Un dels problemes típics quan es treballa amb endomorfismes d'un espai vectorial de dimensió finita és el de determinar bases respecte de les que la matriu de l'endomorfisme és el més simple possible. Per a poder fer aquest estudi hi juga un paper fonamental la determinació dels vectors i valors propis i dels subespais associats. En el paquet **LinearAlgebra** de Maple es poden trobar funcions per a realitzar totes les operacions necessàries.

10.1 Polinomi característic i valors propis

Per a determinar el polinomi característic d'una matriu M podem realitzar els càlculs construint la matriu $M - x\text{Id}$ i demanant el determinat d'aquesta nova matriu

```
> with(LinearAlgebra):
> M:=< <1,1,3> | <2,7,-2> | <1,3,-6> >;
> Mx:=simplify(M-x*IdentityMatrix(3));
> de:=Determinant(Mx);
```

o demanant directament que Maple faci el càlcul i doni directament el polinomi amb la comanda `CharacteristicPolynomial`

```
> CharacteristicPolynomial(M,x);
```

(triar una cosa o l'altre només és una qüestió de gust personal ja que els càlculs que s'han de realitzar internament són essencialment els mateixos).

Pera determinar els valors propis (i la seva multiplicitat) caldrà determinar les arrels del polinomi característic. També es pot fer directament amb la comanda `solve` o amb la comanda específica del paquet **LinearAlgebra** que fa aquesta feina (`Eigenvalues()`).

```
> M1:=< <1,1,3> | <-1,7,-2> | <1,3,-6> >;
> solve(Determinant(M1-x*IdentityMatrix(3)));
> Eigenvalues(M1);
```

Com podreu veure, la diferència entre els dos resultats és ben poca.

10.2 Vectors propis

Un cop es coneixen els valors propis associats a una matriu A , la determinació dels espais de vectors \vec{v} per als que es verifica alguna de les condicions $A\vec{v} = \lambda\vec{v}$ es redueix a la solució de sistemes d'equacions lineals homogenis. La comanda `NullSpace()` del paquet **LinearAlgebra** permet fer la feina.

Exemple 10.1

La matriu

$$A = \begin{pmatrix} -2 & 2 & -3 & -1 \\ 2 & -1 & 2 & 0 \\ 3 & -2 & 4 & 1 \\ -2 & 0 & -2 & -1 \end{pmatrix}$$

té per valors propis 1 i -1 tal i com es pot comprovar amb

```
> A:= <<-2|2|-3|-1>,<2|-1|2|0>,<3|-2|4|1>,<-2|0|-2|-1>>;
> Eigenvalues(A);
```

Els espais de vectors propis amb valor propi 1 o -1 es podran determinar amb

```
> B1:= A-IdentityMatrix(4);
> E1:=NullSpace(B1);
> B2:= A+IdentityMatrix(4);
> E2:=NullSpace(B2);
```

Podreu veure que $E1$ i $E2$ contenen cada una un parell de vectors que generen l'espai de vectors propis corresponent (multipliqueu A per cada una de les columnes que surten en $E1$ o $E2$ per a comprovar que es tenen, efectivament, vectors propis amb valor propi 1 o -1). A la vista dels resultats es pot afirmar que la matriu A és diagonalitzable i es pot confirmar aquest fet considerant

```
> E:=<op(E1)|op(E2)>;
> E^(-1).A.E;
```

Tot el procés de calcular els vectors propis associats a cada un dels valors propis d'una matriu té una comanda específica en el paquet **LinearAlgebra** que es crida amb **Eigenvectors()**. Si considereu

```
> EE:=Eigenvectors(A);
```

obtindreu una seqüència amb dos elements, el primer és una columna amb els valors propis (repetits tants cops com la multiplicitat) i el segon és una matriu que té per columnes les components dels vectors propis ordenades de la mateixa manera que els valors propis (si us hi fixeu, veureu que no surten exactament els mateixos vectors que quan apliqueu **NullSpace()**). Si feu la multiplicació

```
> EE[2]^(-1).A.EE[2];
```

tornareu obtenir la matriu diagonal equivalent a la matriu A .

Amb tot el que heu vist fins ara, l'única dificultat de l'exercici següent hauria de ser la introducció de la matriu M .

Exercici 10.1

Considereu la matriu

$$M = \begin{pmatrix} 1/2 & -1/2 & -1/2 & -1/2 & -1/2 & 0 & 1/2 \\ 7/2 & 9/2 & 3/2 & 7/2 & -3/2 & -1 & -1/2 \\ -1/2 & -5/2 & 1/2 & -1/2 & 3/2 & -2 & 1/2 \\ -5 & -4 & 0 & -6 & 6 & 4 & -1 \\ 0 & 1 & 1 & -2 & 3 & 3 & -1 \\ -7/2 & -9/2 & -1/2 & -11/2 & 11/2 & 3 & -1/2 \\ -3/2 & -3/2 & 5/2 & -11/2 & 17/2 & 4 & -3/2 \end{pmatrix}$$

a) Calculeu el polinomi característic de M .

b) Determineu els seus valors propis i les multiplicitats corresponents.

- c) Determineu els espais de vectors propis associats a cada un dels valors propis.
- d) Comproveu que la matriu M és diagonalitzable.
- e) Calculeu una matriu C tal que $C^{-1} \cdot M \cdot C$ sigui diagonal.

En aquest altre exercici podreu veure una matriu amb el mateix polinomi característic que la matriu M però que no és diagonalitzable.

Exercici 10.2

Considereu

$$B = \begin{pmatrix} 1/2 & -1/2 & -1/2 & -1/2 & -1/2 & 0 & 1/2 \\ 7/2 & 9/2 & 3/2 & 7/2 & -3/2 & -1 & -1/2 \\ -1/2 & -5/2 & 1/2 & -1/2 & 3/2 & -2 & 1/2 \\ -4 & -3 & 1 & -5 & 6 & 3 & -1 \\ 1 & 2 & 2 & -1 & 3 & 2 & -1 \\ -5/2 & -7/2 & 1/2 & -9/2 & 11/2 & 2 & -1/2 \\ 1/2 & 1/2 & 9/2 & -7/2 & 17/2 & 2 & -3/2 \end{pmatrix}$$

calculeu el seu polinomi característic i comproveu que no és diagonalitzable.

Una de les propietats bàsiques del polinomi característic d'una matriu és que *anul·la* a la matriu corresponent. És a dir, si $p(x)$ és el polinomi característic de la matriu A es compleix $p(A) = 0$. En l'exercici següent es posa de manifest aquest fet en un parell d'exemples.

Exercici 10.3

Considereu les matrius A donades per

$$\begin{pmatrix} 1 & 3 & 4 & -1 \\ 0 & 3 & 2 & 1 \\ -1 & -1 & 2 & 5 \\ 2 & 3 & 1 & -1 \end{pmatrix} \quad \begin{pmatrix} -2 & 3 & 1 & -1 \\ 2 & -1 & 0 & 1 \\ -1 & -2 & 2 & 3 \\ 3 & 3 & 1 & -1 \end{pmatrix}$$

calculeu els seus polinomis característics i comproveu que les anul·len.

D'entre tots els polinomis que anul·len un endomorfisme s'en pot determinar el que té el grau més petit (polinomi mínim). La comanda de Maple `MinimalPolynomial()` calcula el polinomi mínim d'una matriu.

Exemple 10.2

Es pot calcular el polinomi mínim de

$$A = \begin{pmatrix} 0 & -1 & 2 \\ 3 & -4 & 6 \\ 2 & -2 & 3 \end{pmatrix}$$

amb

```
> A:= <<0,3,2>|<-1,-4,-2>|<2,6,3> >;
> pm:=MinimalPolynomial(A,x);
> Pm:=unapply(pm,x);
> Pm(A);
```

Observeu que la comanda `Pm` és la funció que s'obté *prenent com a variable* la `x` en l'expressió `pm` (utilitzant `unapply`) que s'aplica correctament sobre una matriu i sobre A dona com a resultat la matriu nul·la.

Exercici 10.4

Calculeu el polinomi característic de la matriu A anterior i comproveu que té els mateixos factors irreductibles que el polinomi mínim (només que amb multiplicitats menors).

10.3 Formes de Jordan

Com ja haureu vist, un endomorfisme d'un espai vectorial (matriu quadrada) no sempre diagonalitza. Per a aquests casos és possible intentar obtenir altres *formes canòniques*. En l'exemple següent veureu una matriu 3×3 que té un únic valor propi i que, a més, l'espai de vectors propis només té dimensió 1.

Exercici 10.5

Sigui

$$A = \begin{pmatrix} 1 & 0 & 1 \\ -1 & 3 & 1 \\ 0 & -1 & 2 \end{pmatrix}.$$

- Calculeu el polinomi característic de A , comproveu que A té un únic valor propi λ (triple) i que l'espai de vectors propis de A té dimensió 1.
- Sigui \vec{v}_3 un vector propi no nul de A . Determineu un vector \vec{v}_2 tal que $(A - \lambda \cdot \text{Id})\vec{v}_2 = \vec{v}_3$ i un vector \vec{v}_1 tal que $(A - \lambda \cdot \text{Id})\vec{v}_1 = \vec{v}_2$.

- c) Comproveu que $\vec{v}_1, \vec{v}_2, \vec{v}_3$ són independents.
- d) Determineu la matriu $P^{-1} \cdot A \cdot P$ si P és la matriu que té per columnes les components dels vectors \vec{v}_i .

La matriu que haureu obtingut en l'últim apartat de l'exercici anterior és el que normalment s'anomena un *bloc de Jordan* (un mateix valor en tots els elements de la diagonal i 1 just en els llocs que estan per sobre, o per sota, d'aquesta diagonal). En general, es poden buscar formes

canòniques de la forma $\left(\begin{array}{c|c|c} J_1 & 0 & \cdots \\ \hline 0 & J_2 & \cdots \\ \hline \vdots & \vdots & \ddots \end{array} \right)$, on cada un dels J_i és un bloc de Jordan.

De la mateixa forma que per als valors i vectors propis, dins del paquet **LinearAlgebra** hi ha comandes específiques per a determinar la forma de Jordan d'un endomorfisme i per a obtenir el canvi de base que permet obtenir aquesta forma a partir de qualsevol altra forma equivalent. La comanda que fa les dues feines és `JordanForm()`. Per a obtenir la forma de Jordan equivalent a una matriu A donada basta fer `JordanForm(A)`; si es vol la matriu de canvi es pot modificar la comanda anterior i posar `JordanForm(A, output='Q')`; i si es volen les dues coses `JordanForm(A, output=['J', 'Q'])`; Veieu a continuació uns exemples

Exemple 10.3

```
> A:=<<2,2,0,0>|<1/2,3/2,-1/2,1/2>|<0,0,2,0>|<-1/2,-3/2,1/2,3/2>>;
> JordanForm(A);
> B:= JordanForm(A,output='Q');
> C:= JordanForm(A,output=['J', 'Q']);
> simplify(B^(-1).A.B);
> simplify(C[2]^(-1).A.C[2]);
```

No ha de ser difícil fer l'exercici següent:

Exercici 10.6

Considereu la matriu A donada per

$$\begin{pmatrix} 3 & 0 & 0 & 1 & 4 & -3 & 2 \\ -1 & 3 & 0 & -2 & -5 & 4 & -2 \\ 0 & 2 & 3 & -3 & -1 & 2 & 1 \\ 0 & 1 & 0 & 1 & -1 & 1 & 0 \\ -2 & -2 & -1 & 0 & -8 & 6 & -7 \\ -1 & -2 & -1 & 1 & -6 & 5 & -5 \\ 2 & 0 & 1 & 1 & 9 & -6 & 7 \end{pmatrix}$$

i la matriu B

$$\begin{pmatrix} 2 & 1 & -1 & 1 & 4 & -3 & 3 \\ 0 & 3 & 0 & -2 & -4 & 3 & -2 \\ 0 & 3 & 2 & -3 & 0 & 1 & 2 \\ 0 & 1 & 0 & 1 & -1 & 1 & 0 \\ 0 & -4 & 1 & 0 & -8 & 6 & -9 \\ 0 & -4 & 1 & 1 & -7 & 6 & -7 \\ 0 & 1 & 0 & 1 & 8 & -5 & 8 \end{pmatrix}$$

Existeix alguna matriu invertible P tal que $P^{-1} \cdot A \cdot P = B$? (Determineu les formes de Jordan de A i B).

Exercici 10.7

Comproveu que existeix una matriu Q (com a mínim) tal que $Q^{-1} \cdot A \cdot Q = B$ per a

$$A = \begin{pmatrix} 5/2 & -1/2 & -1/2 & 0 & 0 \\ -5 & -1 & -1 & 1 & -1 \\ 11/2 & 5/2 & 5/2 & -1 & 1 \\ 2 & 0 & 0 & 1 & 0 \\ 6 & 4 & 2 & -1 & 3 \end{pmatrix} \quad B = \begin{pmatrix} 1 & 0 & 0 & -1 & 0 \\ 1 & 1 & -1 & 1 & 1 \\ 3 & -1 & -1 & 2 & 3 \\ 0 & 1 & 1 & 2 & -1 \\ 3 & -2 & -3 & 2 & 5 \end{pmatrix}$$

i calculeu una d'aquestes matrius Q .

11 Càlculs de forma interactiva (*maplets*)

Una de les novetats del programa Maple en les versions 8 i 9 respecte de les versions anteriors és la possibilitat d'utilitzar i definir procediments per a treballar de forma interactiva utilitzant finestres, quadres de diàleg, llistes d'opcions, menús, ... No tractarem, de moment, com escriure procediments de tipus *maplet* però hi ha alguns dels paquets estàndard que contenen alguns exemples d'aquest tipus de procediments. En particular, i si es tracta de resoldre problemes d'àlgebra lineal, trobarem en el paquet **Student[LinearAlgebra]**¹ les comandes

```
EigenvaluesTutor  EigenvectorsTutor  GaussianEliminationTutor
GaussJordanEliminationTutor  InverseTutor  LinearSolveTutor
```

Veiem amb alguns exemples el funcionament d'aquestes funcions. Carregueu en primer lloc el paquet amb la comanda

```
> with(Student[LinearAlgebra]);
```

11.1 GaussianEliminationTutor

La comanda **GaussianEliminationTutor** redueix interactivament una matriu (de mides fins a 5×5) fins arribar a una forma esglaonada. Un cop s'ha arribat al resultat, es poden obtenir el rang i la dimensió del nucli, una base de l'espai que generen les files, una del que generen les columnes i una del nucli.

Exemple 11.1

Es pot observar que la matriu A donada per

$$A = \begin{pmatrix} -1 & 2 & 3 & 0 & 1 \\ 0 & 1 & 1 & 2 & 3 \\ 0 & 0 & 0 & 1 & 1 \\ 1 & -1 & -2 & 0 & 0 \\ -2 & 2 & 4 & -1 & -1 \end{pmatrix}$$

té rang 3 i realitzar els càlculs de reducció de forma interactiva fent

```
> A:= <<-1|2|3|0|1>, <0|1|1|2|3>, <0|0|0|1|1>, <1|-1|-2|0|0>, <-2|2|4|-1|-1>>;
> GaussianEliminationTutor(A);
```

Veureu que, a part d'oferir botons i desplegable per a seleccionar les operacions que es volen realitzar la mateixa aplicació pot donar indicacions sobre el camí que es pot seguir per a fer la reducció i que acaba donant com a resultat final la matriu reduïda que s'hagi calculat.

¹Aquest paquet és, de fet, un sub-paquet del paquet **Student** el qual proporciona funcions que serveixen per ajudar a explicar i a estudiar alguns dels temes matemàtics bàsics.

11.2 InverseTutor

Es pot fer el càlcul de la inversa d'una matriu (de mides fins a 5×5) pas a pas utilitzant la comanda `InverseTutor` com en l'exemple següent.

Exemple 11.2

```
> M:=<<1|2|3>,<2|-1|0>,<0|1|-1>>;  
> InverseTutor(M);
```

Com en el cas anterior es pot fer que el resultat final que doni Maple sigui la matriu inversa que s'ha calculat.

11.3 EigenvectorsTutor

Aquest *maplet* ensenya els càlculs que es van fer per a calcular els valors i vectors propis d'una matriu (de mides fins a 4×4) i acaba donant com a resultat un conjunt amb la base de vectors propis. Podeu provar el seu funcionament en l'exemple següent.

Exemple 11.3

```
> M:=<<0,-2,-4,4>|<8,22,16,-28>|<-4,-10,-6,14>|<4,10,8,-12>>;  
> EigenvectorsTutor(M);
```

No us oblideu, finalment, de fer un cop d'ull a la secció d'ajuda de Maple sobre aquestes comandes.